A Computer-Aided Proof of a Corrected Version of 10.2.32 in Abramowitz & Stegun's HMS

$$\left(\frac{\partial}{\partial \nu} I_\nu(x)\right)\Bigg|_{\nu=\frac{1}{2}} = -\frac{E_1(2\,x)\,\mathbf{e}^{x} + \mathrm{Ei}(2\,x)\,\mathbf{e}^{(-x)}}{\sqrt{2\,\pi\,x}}$$

Frédéric Chyzak (June 10, 2005)

This session has been prepared with Maple9.5 and no additional library.

For simplicity of treatment, we restrict ourselves to positive real x. Discussing the identity on other domains requires handling branch cuts; see the concluding remarks. Note that for $0 < x$ and in traditional notation (A&S), $E_1(x) = -\mathrm{Ei}(-x)$. This also equals the value $\mathrm{Ei}(1, x)$ in Maple notation. The right-hand side of the identity under study therefore is:

```
> fo:=-1/sqrt(2)/sqrt(Pi)/sqrt(x)*(Ei(2*x)*exp(-x)-Ei(-2*x)*exp(x));
```

$$fo := -\frac{\sqrt{2}\,(\mathrm{Ei}(2\,x)\,\mathbf{e}^{(-x)} - \mathrm{Ei}(-2\,x)\,\mathbf{e}^{x})}{2\,\sqrt{\pi}\,\sqrt{x}}$$

Determining a Fourth-Order ODE Satisfied by the Left-Hand Side, and its General Solution

Start with the ODE satisfied by $I_\nu(x)$.

```
> deq1:=x^2*diff(y(x),x,x)+x*diff(y(x),x)-(x^2+nu^2)*y(x);
```

$$deq1 := x^2\left(\frac{d^2}{dx^2}\,y(x)\right) + x\left(\frac{d}{dx}\,y(x)\right) - (x^2+\nu^2)\,y(x)$$

```
> dsolve(deq1,y(x));
```

$$y(x) = \_C1\,\mathrm{BesselI}(\nu, x) + \_C2\,\mathrm{BesselK}(\nu, x)$$

Taking derivative with respect to ν allows us to get an inhomogeneous ODE in $\frac{\partial}{\partial \nu} I_\nu(x)$ with right-hand side in terms of $I_\nu(x)$.

```
> deq2:=diff(deq1,nu)+eval(deq1,y(x)=z(x));
```

$$deq2 := -2\,\nu\, y(x) + x^2 \left(\frac{d^2}{dx^2} z(x)\right) + x\left(\frac{d}{dx} z(x)\right) - (x^2+\nu^2)\, z(x)$$

The next step is a (differential) elimination of $y(x)$ (and possibly of its derivatives) between *deq1* and *deq2*. To this end, an annihilating operator for the term in *deq2* that involves $y(x)$ can in the general situation be obtained by existing code (Mgfun[dfinite_expr_to_diffeq]). Here, this term is so simple that no computation is required. Upon applying its annihilating operator, we get a homogeneous ODE in $\frac{\partial}{\partial \nu} I_\nu(x)$.

```
> collect(eval(deq1,y(x)=deq2)+2*nu*deq1,{z,diff},factor);
```

$$-x\,(6x^2-1+2\nu^2)\left(\frac{d}{dx} z(x)\right) - x^2(-7+2x^2+2\nu^2)\left(\frac{d^2}{dx^2} z(x)\right) + 6x^3\left(\frac{d^3}{dx^3} z(x)\right) + x^4\left(\frac{d^4}{dx^4} z(x)\right) + (\nu^2+2x+x^2)(\nu^2-2x+x^2)\, z(x)$$

The relation to be obtained is for $\nu = \frac{1}{2}$, and the ODE becomes:

```
> deq:=collect(16*eval(%,nu=1/2),{z,diff},factor);
```

$$deq := -8\,x\,(12x^2-1)\left(\frac{d}{dx} z(x)\right) - 8x^2(-13+4x^2)\left(\frac{d^2}{dx^2} z(x)\right) + 96x^3\left(\frac{d^3}{dx^3} z(x)\right) + 16x^4\left(\frac{d^4}{dx^4} z(x)\right) + (1+8x+4x^2)(1-8x+4x^2)\, z(x)$$

We easily check that $\left.\left(\frac{\partial}{\partial \nu} I_\nu(x)\right)\right|_{\nu=\frac{1}{2}}$ is a solution to this equation: by inserting $I_\nu(x)$ into it, we get a double zero at $\nu = \frac{1}{2}$.

```
> factor(simplify(eval(deq,z(x)=BesselI(nu,x))));
```

$$\mathrm{BesselI}(\nu, x)\,(2\nu-1)^2\,(2\nu+1)^2$$

At this point, we readily verify that the right-hand side of BS.37.6 is also a solution of the differential equation *deq*.

```
> simplify(eval(deq,z(x)=fo));
```

$$0$$

Better than just verifying the identity, we want to derive it by computing the conditions. The general solution of the ODE is

```
> dsolve(deq);
```

$$z(x) = \frac{\_C1\, e^{x}}{\sqrt{x}} + \frac{\_C2\, e^{(-x)}}{\sqrt{x}} + \frac{\_C3\, e^{x}\, \mathrm{Ei}(1, 2\, x)}{\sqrt{x}} + \frac{\_C4\, e^{(-x)}\, \mathrm{Ei}(1, -2\, x)}{\sqrt{x}}$$

Here, for positive x, we have both relations $\mathrm{Ei}(1, x) = -\mathrm{Ei}(-x)$ and $\mathrm{Ei}(1, -x) = -\mathrm{Ei}(x) - I\, \pi$, so that, up to a change of the constants in the general solution above, we may replace it with

```
> dsol:=_C1*exp(x)/sqrt(x)+_C2*exp(-x)/sqrt(x)+_C3*Ei(-2*x)*exp(x)/sqr
  (x)+_C4*Ei(2*x)*exp(-x)/sqrt(x);
```

$$dsol := \frac{\_C1\, e^{x}}{\sqrt{x}} + \frac{\_C2\, e^{(-x)}}{\sqrt{x}} + \frac{\_C3\, \mathrm{Ei}(-2\, x)\, e^{x}}{\sqrt{x}} + \frac{\_C4\, \mathrm{Ei}(2\, x)\, e^{(-x)}}{\sqrt{x}}$$

Summarizing, we have obtained the following fourth-order ODE satisfied by $\left.\left(\frac{\partial}{\partial \nu} I_{\nu}(x)\right)\right|_{\nu = \frac{1}{2}}$ as well as its general solution:

```
> deq;
```

$$-8\, x\, (12\, x^2 - 1) \left(\frac{d}{dx} z(x)\right) - 8\, x^2\, (-13 + 4\, x^2) \left(\frac{d^2}{dx^2} z(x)\right) + 96\, x^3 \left(\frac{d^3}{dx^3} z(x)\right) + 16\, x^4 \left(\frac{d^4}{dx^4} z(x)\right) + (1 + 8\, x + 4\, x^2)\, (1 - 8\, x + 4\, x^2)\, z(x)$$

```
> dsol;
```

$$\frac{\_C1\, e^{x}}{\sqrt{x}} + \frac{\_C2\, e^{(-x)}}{\sqrt{x}} + \frac{\_C3\, \mathrm{Ei}(-2\, x)\, e^{x}}{\sqrt{x}} + \frac{\_C4\, \mathrm{Ei}(2\, x)\, e^{(-x)}}{\sqrt{x}}$$

Identifying the Right-Hand Side as a Specific Solution of the Fourth-Order ODE

In the theory, "initial conditions" are really first terms of asymptotic expansions. We proceed to set up a linear system to determine the constants $\_C1$, ..., above.

```
> Order:=3;
```

$$Order := 3$$

In order to identify $\left.\left(\frac{\partial}{\partial \nu} I_{\nu}(x)\right)\right|_{\nu=\frac{1}{2}}$, we will use the MultiSeries package (Meunier, Salvy, and Sedoglavic), which deals with general asymptotic scales and parameterized asymptotics.

```
> with(MultiSeries,series);
```

Warning, the protected name series has been redefined and unprotected

$$[series]$$

```
> series(BesselI(nu,x),x);
```

$$\frac{x^{\nu}}{\Gamma(\nu+1)\,\mathbf{e}^{(\ln(2)\,\nu)}}+\frac{x^{(\nu+2)}}{4\,(\nu+1)\,\Gamma(\nu+1)\,\mathbf{e}^{(\ln(2)\,\nu)}}+O(x^{(4+\nu)})$$

```
> diff(%,nu);
```

$$-\frac{x^{\nu}\,\Psi(\nu+1)}{\Gamma(\nu+1)\,\mathbf{e}^{(\ln(2)\,\nu)}}-\frac{x^{\nu}\ln(2)}{\Gamma(\nu+1)\,\mathbf{e}^{(\ln(2)\,\nu)}}+\frac{x^{\nu}\ln(x)}{\Gamma(\nu+1)\,\mathbf{e}^{(\ln(2)\,\nu)}}-\frac{x^{(\nu+2)}}{4\,(\nu+1)^{2}\,\Gamma(\nu+1)\,\mathbf{e}^{(\ln(2)\,\nu)}}-\frac{x^{(\nu+2)}\,\Psi(\nu+1)}{4\,(\nu+1)\,\Gamma(\nu+1)\,\mathbf{e}^{(\ln(2)\,\nu)}}-\frac{x^{(\nu+2)}\ln(2)}{4\,(\nu+1)\,\Gamma(\nu+1)\,\mathbf{e}^{(\ln(2)\,\nu)}}+\frac{x^{(\nu+2)}\ln(x)}{4\,(\nu+1)\,\Gamma(\nu+1)\,\mathbf{e}^{(\ln(2)\,\nu)}}+O(x^{(4+\nu)}\ln(x))$$

A truncated asymptotic expansion for $\left.\left(\frac{\partial}{\partial \nu} I_{\nu}(x)\right)\right|_{\nu=\frac{1}{2}}$ therefore is:

```
> fs:=map(simplify,eval(%,nu=1/2));
```

$$fs:=\frac{\sqrt{2}\,\sqrt{x}\,(-2+\gamma+2\ln(2))}{\sqrt{\pi}}-\frac{\sqrt{2}\,\sqrt{x}\ln(2)}{\sqrt{\pi}}+\frac{\sqrt{2}\,\sqrt{x}\ln(x)}{\sqrt{\pi}}-\frac{\sqrt{2}\,x^{(5/2)}}{9\sqrt{\pi}}+\frac{\sqrt{2}\,x^{(5/2)}\,(-2+\gamma+2\ln(2))}{6\sqrt{\pi}}-\frac{\sqrt{2}\,x^{(5/2)}\ln(2)}{6\sqrt{\pi}}+\frac{\sqrt{2}\,x^{(5/2)}\ln(x)}{6\sqrt{\pi}}+O(x^{(9/2)}\ln(x))$$

Returning to the general solution to the 4th-order ODE, we derive a truncated asymptotic expansion for it:

```
> series(dsol,x);
```

$$\frac{1}{\sqrt{x}}\left(\left(\frac{\_C1}{\ln\left(\frac{1}{x}\right)}+\frac{\_C2}{\ln\left(\frac{1}{x}\right)}+\frac{\_C3\,\gamma}{\ln\left(\frac{1}{x}\right)}-\_C3+\frac{\_C3\ln(2)}{\ln\left(\frac{1}{x}\right)}+\frac{\_C4\,\gamma}{\ln\left(\frac{1}{x}\right)}-\_C4+\frac{\_C4\ln(2)}{\ln\left(\frac{1}{x}\right)}\right)\ln\left(\frac{1}{x}\right)\right)-\left(-\frac{\_C1}{\ln\left(\frac{1}{x}\right)}+\frac{\_C2}{\ln\left(\frac{1}{x}\right)}-\frac{\_C3\,\gamma}{\ln\left(\frac{1}{x}\right)}+\_C3-\frac{\_C3\ln(2)}{\ln\left(\frac{1}{x}\right)}+\frac{2\,\_C3}{\ln\left(\frac{1}{x}\right)}+\frac{\_C4\,\gamma}{\ln\left(\frac{1}{x}\right)}-\_C4+\frac{\_C4\ln(2)}{\ln\left(\frac{1}{x}\right)}-\frac{2\,\_C4}{\ln\left(\frac{1}{x}\right)}\right)\ln\left(\frac{1}{x}\right)\sqrt{x}+\frac{1}{2}\left(\frac{\_C1}{\ln\left(\frac{1}{x}\right)}+\frac{\_C2}{\ln\left(\frac{1}{x}\right)}+\frac{\_C3\,\gamma}{\ln\left(\frac{1}{x}\right)}-\_C3+\frac{\_C3\ln(2)}{\ln\left(\frac{1}{x}\right)}-\frac{2\,\_C3}{\ln\left(\frac{1}{x}\right)}+\frac{\_C4\,\gamma}{\ln\left(\frac{1}{x}\right)}-\_C4+\frac{\_C4\ln(2)}{\ln\left(\frac{1}{x}\right)}-\frac{2\,\_C4}{\ln\left(\frac{1}{x}\right)}\right)\ln\left(\frac{1}{x}\right)x^{(3/2)}+O(x^{(5/2)})$$

```
> dsol_fs:=eval(%,nu=1/2);
```

$$dsol\_fs := \frac{1}{\sqrt{x}}\left(\left(\frac{\_C1}{\ln\left(\frac{1}{x}\right)}+\frac{\_C2}{\ln\left(\frac{1}{x}\right)}+\frac{\_C3\,\gamma}{\ln\left(\frac{1}{x}\right)}-\_C3+\frac{\_C3\ln(2)}{\ln\left(\frac{1}{x}\right)}+\frac{\_C4\,\gamma}{\ln\left(\frac{1}{x}\right)}-\_C4+\frac{\_C4\ln(2)}{\ln\left(\frac{1}{x}\right)}\right)\ln\left(\frac{1}{x}\right)\right)-\left(-\frac{\_C1}{\ln\left(\frac{1}{x}\right)}+\frac{\_C2}{\ln\left(\frac{1}{x}\right)}-\frac{\_C3\,\gamma}{\ln\left(\frac{1}{x}\right)}+\_C3-\frac{\_C3\ln(2)}{\ln\left(\frac{1}{x}\right)}+\frac{2\,\_C3}{\ln\left(\frac{1}{x}\right)}+\frac{\_C4\,\gamma}{\ln\left(\frac{1}{x}\right)}-\_C4+\frac{\_C4\ln(2)}{\ln\left(\frac{1}{x}\right)}-\frac{2\,\_C4}{\ln\left(\frac{1}{x}\right)}\right)\ln\left(\frac{1}{x}\right)\sqrt{x}+\frac{1}{2}\left(\frac{\_C1}{\ln\left(\frac{1}{x}\right)}+\frac{\_C2}{\ln\left(\frac{1}{x}\right)}+\frac{\_C3\,\gamma}{\ln\left(\frac{1}{x}\right)}-\_C3+\frac{\_C3\ln(2)}{\ln\left(\frac{1}{x}\right)}-\frac{2\,\_C3}{\ln\left(\frac{1}{x}\right)}+\frac{\_C4\,\gamma}{\ln\left(\frac{1}{x}\right)}-\_C4+\frac{\_C4\ln(2)}{\ln\left(\frac{1}{x}\right)}-\frac{2\,\_C4}{\ln\left(\frac{1}{x}\right)}\right)\ln\left(\frac{1}{x}\right)x^{(3/2)}+O(x^{(5/2)})$$

Identifying the known solution with this general solution yields a linear system

```
> zz:=convert(simplify(2*sqrt(Pi)*sqrt(x)*series(dsol_fs-fs,x)),polyno
);
```

$$zz := 2\sqrt{\pi}\,\_C1+2\sqrt{\pi}\,\_C2-2x\sqrt{\pi}\,\_C2-4x\sqrt{\pi}\,\_C3+4x\sqrt{\pi}\,\_C4-2x\sqrt{2}+4x\sqrt{2}+2\sqrt{\pi}\,\_C4\,\gamma+2\sqrt{\pi}\,\_C4\ln(2)+2\sqrt{\pi}\,\_C3\,\gamma+2\sqrt{\pi}\,\_C3\ln(2)$$

$$+\,2\,x\,\sqrt{\pi}\,\_C3\,\gamma+2\,x\,\sqrt{\pi}\,\_C3\,\ln(2)-2\,x\,\sqrt{\pi}\,\_C3\,\ln\left(\frac{1}{x}\right)-2\,x\,\sqrt{\pi}\,\_C4\,\gamma$$
$$+\,2\,x\,\sqrt{\pi}\,\_C4\,\ln\left(\frac{1}{x}\right)-2\,x\,\sqrt{\pi}\,\_C4\,\ln(2)+x^2\,\sqrt{\pi}\,\_C3\,\gamma-x^2\,\sqrt{\pi}\,\_C3\,\ln\left(\frac{1}{x}\right)$$
$$+\,x^2\,\sqrt{\pi}\,\_C3\,\ln(2)+x^2\,\sqrt{\pi}\,\_C4\,\gamma-x^2\,\sqrt{\pi}\,\_C4\,\ln\left(\frac{1}{x}\right)+x^2\,\sqrt{\pi}\,\_C4\,\ln(2)$$
$$-\,2\,x^2\,\sqrt{\pi}\,\_C4-2\,\sqrt{\pi}\,\_C3\,\ln\left(\frac{1}{x}\right)-2\,\sqrt{\pi}\,\_C4\,\ln\left(\frac{1}{x}\right)-2\,x\,\ln(2)\,\sqrt{2}$$
$$+\,2\,x\,\sqrt{2}\,\ln\left(\frac{1}{x}\right)+x^2\,\sqrt{\pi}\,\_C1+x^2\,\sqrt{\pi}\,\_C2-2\,x^2\,\sqrt{\pi}\,\_C3+2\,x\,\sqrt{\pi}\,\_C1$$

```
> sys:={coeffs(zz,{x,ln(1/x)})};
```

$$sys:=\{-2\,\sqrt{\pi}\,\_C3-2\,\sqrt{\pi}\,\_C4,\ -\sqrt{\pi}\,\_C4-\sqrt{\pi}\,\_C3,\ 2\,\sqrt{\pi}\,\_C4+2\,\sqrt{2}-2\,\sqrt{\pi}\,\_C3,\ 2\,\sqrt{\pi}\,\_C1+2\,\sqrt{\pi}\,\_C2+2\,\sqrt{\pi}\,\_C3\,\ln(2)+2\,\sqrt{\pi}\,\_C4\,\gamma+2\,\sqrt{\pi}\,\_C4\,\ln(2)+2\,\sqrt{\pi}\,\_C3\,\gamma,\ -2\,\sqrt{\pi}\,\_C2-4\,\sqrt{\pi}\,\_C3+4\,\sqrt{\pi}\,\_C4-2\,\sqrt{2}\,\gamma+4\,\sqrt{2}+2\,\sqrt{\pi}\,\_C3\,\gamma+2\,\sqrt{\pi}\,\_C3\,\ln(2)-2\,\ln(2)\,\sqrt{2}-2\,\sqrt{\pi}\,\_C4\,\gamma-2\,\sqrt{\pi}\,\_C4\,\ln(2)+2\,\sqrt{\pi}\,\_C1,\ \sqrt{\pi}\,\_C4\,\gamma+\sqrt{\pi}\,\_C3\,\gamma+\sqrt{\pi}\,\_C4\,\ln(2)+\sqrt{\pi}\,\_C3\,\ln(2)+\sqrt{\pi}\,\_C1+\sqrt{\pi}\,\_C2-2\,\sqrt{\pi}\,\_C3-2\,\sqrt{\pi}\,\_C4\}$$

```
> sol:=solve(sys,{_C1,_C2,_C3,_C4});
```

$$sol:=\left\{\_C2=0,\ \_C1=0,\ \_C4=-\frac{\sqrt{2}}{2\,\sqrt{\pi}},\ \_C3=\frac{\sqrt{2}}{2\,\sqrt{\pi}}\right\}$$

Our final expression for $\left.\left(\frac{\partial}{\partial\nu}I_\nu(x)\right)\right|_{\nu=\frac{1}{2}}$ is

```
> expr:=eval(dsol,sol);
```

$$expr:=\frac{\sqrt{2}\,\mathrm{Ei}(-2\,x)\,\mathbf{e}^{x}}{2\,\sqrt{\pi}\,\sqrt{x}}-\frac{\sqrt{2}\,\mathrm{Ei}(2\,x)\,\mathbf{e}^{(-x)}}{2\,\sqrt{\pi}\,\sqrt{x}}$$

It is really the announced right-hand side.

```
> normal(expr-fo);
```

$$0$$

(Repeated:)

```
> expr;
```

$$\frac{\sqrt{2}\,\mathrm{Ei}(-2\,x)\,\mathbf{e}^{x}}{2\,\sqrt{\pi}\,\sqrt{x}}-\frac{\sqrt{2}\,\mathrm{Ei}(2\,x)\,\mathbf{e}^{(-x)}}{2\,\sqrt{\pi}\,\sqrt{x}}$$

Beyond Branch Cuts

The exponential integrals and logarithms implied in all asymptotic expansions induce branch cuts on the positive real line and negative real line. Of course, the derivation above could be reproduced with adequate changes for $x < 0$, $0 < \Im(x)$, and $\Im(x) < 0$, respectively.

A uniform treatment of the identity and functions above for a complex argument, taking branch cuts into account, is beyond this session and beyond the current implementations. The topic is the core of Ludovic Meunier's thesis (in preparation).

```
>
```